
1. Overview

This document lists all changes needed to correct problems in the TTCN implementation of Rel-13 eMTC test case 6.1.2.2c which is part of the LTE-A PRO test suite in the ‘iwd-TTCN3-B2016-06_D16wk49’ ATS delivery.

The test case can be demonstrated to run with 3x different Rel-13 LTE-A PRO UEs (see section 5). Execution logs are provided as evidence.

3. Verification Test Summary

Test Case:
TC_6_1_2_2c
Test Group:
LTE_A_PRO\6\Idle_CellReselection_eMTC.ttcn
ATS Version:
iwd-TTCN3-B2016-06_D16wk49
System Simulator used:
R&S CMW500
UE used:
Altair ALT1210, Qualcomm MDM 9607, Sequans Monarch LTE Cat M1

Verification Status:
PASS


4. Corrections required for TC 6.1.2.2c
Change 1 – Common corrections for eMTC test cases
Change 1.1 – Correction to SI-BR scheduling
	Function name
	f_EUTRA_SetSI_BRSubframeOffsetList ()

	Reason for change
	SI-BRs in eMTC do not have DCI, so the Redundancy Version of the scheduled data depends on the subframe number. In the official code (and TS 36.523-3 spec), SI-BRs are scheduled in subframe#2, which would mean Redundancy Version = 3. But this is not self-decodable, so the UE won't be able to decode any SI-BRs.

	Summary of change
	Use RV=0 by scheduling the SI-BRs in subframe#0 (MIB on subframe#0 in the middle of the Band does not collide with them).

	TTCN module
	Common\EUTRA\EUTRA_ConfigurationSteps.ttcn

	MCC160 Comment
	


Before change

...

  function f_EUTRA_SetSI_BRSubframeOffsetList(SystemInformationBlockType1_BR_r13 p_SIB1_BR,
                                              EUTRA_FDD_TDD_Mode_Type p_FDD_TDD)
    return template (value) SI_SubframeOffsetList_Type
  {
    var template (value) SI_SubframeOffsetList_Type v_SI_SubframeOffsetList;
    var integer v_NoOfSIs := lengthof(p_SIB1_BR.schedulingInfoList);
    var integer v_SubframeOffset;
    var integer i;
    //Acc to 36.523-3 Table 7.7.2a-1
    for (i := 0; i < v_NoOfSIs; i := i+1) {
      /* Assigne the subframe offset according to 36.523-3 Table 7.7.2a-1;
         => there is a maximum of 4 SIs */
      select (p_FDD_TDD) {
        case (FDD) {
          select (i) {
            case (0, 1, 2, 3) {   // Subframe offset for SI1-BR, SI2-BR, SI13-BR, SI14-BR
              v_SubframeOffset := 2;
            }
            case else {
              FatalError (__FILE__, __LINE__, "No subframe offset defined for that SI-BR");
            }
          }
        }
        case (TDD) {
          select (i) {
            case (0, 1, 2, 3) { // Subframe offset for SI1-BR, SI2-BR, SI13-BR, SI14-BR
              v_SubframeOffset := 0;
            }
            case else {
              FatalError (__FILE__, __LINE__, "No subframe offset defined for that SI-BR");
            }
          }
        }
      }
      v_SI_SubframeOffsetList[i] := v_SubframeOffset;
    }
    return v_SI_SubframeOffsetList;
  }
...

After change

...

  function f_EUTRA_SetSI_BRSubframeOffsetList(SystemInformationBlockType1_BR_r13 p_SIB1_BR,
                                              EUTRA_FDD_TDD_Mode_Type p_FDD_TDD)
    return template (value) SI_SubframeOffsetList_Type
  {
    var template (value) SI_SubframeOffsetList_Type v_SI_SubframeOffsetList;
    var integer v_NoOfSIs := lengthof(p_SIB1_BR.schedulingInfoList);
    var integer v_SubframeOffset;
    var integer i;
    //Acc to 36.523-3 Table 7.7.2a-1
    for (i := 0; i < v_NoOfSIs; i := i+1) {
      /* Assigne the subframe offset according to 36.523-3 Table 7.7.2a-1;
         => there is a maximum of 4 SIs */
      select (p_FDD_TDD) {
        case (FDD) {
          select (i) {
            case (0, 1, 2, 3) {   // Subframe offset for SI1-BR, SI2-BR, SI13-BR, SI14-BR
              v_SubframeOffset := 0;
            }
            case else {
              FatalError (__FILE__, __LINE__, "No subframe offset defined for that SI-BR");
            }
          }
        }
        case (TDD) {
          select (i) {
            case (0, 1, 2, 3) { // Subframe offset for SI1-BR, SI2-BR, SI13-BR, SI14-BR
              v_SubframeOffset := 0;
            }
            case else {
              FatalError (__FILE__, __LINE__, "No subframe offset defined for that SI-BR");
            }
          }
        }
      }
      v_SI_SubframeOffsetList[i] := v_SubframeOffset;
    }
    return v_SI_SubframeOffsetList;
  }
...

Change 1.2 – Correction to dedicated SRS configuration
	Function name
	fl_EUTRA_InitFDD_TDD_FromBand ()

	Reason for change
	The current Dedicated SRS configuration is not applicable for eMTC, e.g. the Bandwidth (bw0) is far too big.

	Summary of change
	Analog to the basic FDD-HD mode, disable the UL SRS for eMTC.

	TTCN module
	Common\EUTRA\EUTRA_CellInfo.ttcn

	MCC160 Comment
	


Before change

...

  function fl_EUTRA_InitFDD_TDD_FromBand(integer p_FrequencyBand) return EUTRA_FDD_TDD_CellInfo_Type
  { //@sic R5-163085 p_FrequencyBand type changed to integer sic@
    var template (value) EUTRA_FDD_TDD_CellInfo_Type v_FDD_TDD_Info;
    if (f_EUTRA_FrequencyBandIsFDD(p_FrequencyBand)) {              // Bands below 33 are reserved for FDD
      v_FDD_TDD_Info := cs_EUTRA_FDD_Info( pc_FDD_TypeB_HalfDuplex ); // @sic R5s160586 sic@
    } else {                                 // Bands above 33 are reserved for TDD
      v_FDD_TDD_Info := cs_EUTRA_TDD_Info(px_eTDDsubframeConfig);  //@sic R5-096691  sic@
    }
    return valueof(v_FDD_TDD_Info);          // Note: valueof cannot be avoided here
  }
...

After change

...

  function fl_EUTRA_InitFDD_TDD_FromBand(integer p_FrequencyBand) return EUTRA_FDD_TDD_CellInfo_Type
  { //@sic R5-163085 p_FrequencyBand type changed to integer sic@
    var template (value) EUTRA_FDD_TDD_CellInfo_Type v_FDD_TDD_Info;
    if (f_EUTRA_FrequencyBandIsFDD(p_FrequencyBand)) {              // Bands below 33 are reserved for FDD
      v_FDD_TDD_Info := cs_EUTRA_FDD_Info( pc_FDD_TypeB_HalfDuplex or pc_ue_CategoryDL_M1 or f_GetTestcaseAttrib_EnhancedCoverage(testcasename()) ); // @sic R5s160586 sic@
    } else {                                 // Bands above 33 are reserved for TDD
      v_FDD_TDD_Info := cs_EUTRA_TDD_Info(px_eTDDsubframeConfig);  //@sic R5-096691  sic@
    }
    return valueof(v_FDD_TDD_Info);          // Note: valueof cannot be avoided here
  }
...

Change 1.3 – Correction to initialization of the MPDCCH Narrowbands to monitor
	Function name
	f_EUTRA_ReturnMpdcchPNB ()

	Reason for change
	TTCN initializes the parameters of all defined Cell IDs in the beginning of the TC execution and so the switch case n50 goes always to fatal error in the function allowing only certain Cell IDs. Additionally, the configured values do not match the specification in TS 36.523-3, since field values (1..maxAvailNarrowBands-r13) correspond to narrowband indices (0..[maxAvailNarrowBands-r13-1]) as specified in TS 36.211 / TS 36.331.

	Summary of change
	Removing Cell ID check from the switch case n50 and correcting the values of MPDCCH Narrowbands to monitor.

	TTCN module
	Common\EUTRA\EUTRA_CellInfo.ttcn

	MCC160 Comment
	


Before change

...

  function f_EUTRA_ReturnMpdcchPNB(EUTRA_CellId_Type p_CellId,
                                   Dl_Bandwidth_Type p_DL_ChBandwidth,
                                   SysinfoType_Type p_SysinfoType)
    return Mpdcch_NarrowbandsToMonitor_r13_Type
  {
    var Mpdcch_NarrowbandsToMonitor_r13_Type v_Mpdcch_NarrowbandsToMonitor;
    if (p_SysinfoType == sysinfoAndBR) { // Initiatile parameters for CE Mode A or B
      select(p_DL_ChBandwidth) {
        case (n25) { //5MHz
          // 36.523-3 Table 7.3.3.9.1-2: Narrow band allocation per cell (5 MHz)
          v_Mpdcch_NarrowbandsToMonitor := {1}
        }
        case (n50) { //10MHz
          //36.523-3 Table 7.3.3.9.1-4: Narrow band allocation per cell (10 MHz)
          select (p_CellId) {

            case (eutra_Cell1, eutra_Cell2) { v_Mpdcch_NarrowbandsToMonitor := {1} }

            case (eutra_Cell4) { v_Mpdcch_NarrowbandsToMonitor := {2} }

            case else { FatalError(__FILE__, __LINE__, "invalid cellId for eMTC");}

          }
        }
        case (n100) {//20MHz
          //36.523-3 Table 7.3.3.9.1-6: Narrow band allocation per cell (20 MHz)
          v_Mpdcch_NarrowbandsToMonitor := {7}
        }
        case else { FatalError(__FILE__, __LINE__, "invalid value for ChannelBandwidth");}
      }
    }
    return v_Mpdcch_NarrowbandsToMonitor;
  }
...

After change

...

  function f_EUTRA_ReturnMpdcchPNB(EUTRA_CellId_Type p_CellId,
                                   Dl_Bandwidth_Type p_DL_ChBandwidth,
                                   SysinfoType_Type p_SysinfoType)
    return Mpdcch_NarrowbandsToMonitor_r13_Type
  {
    var Mpdcch_NarrowbandsToMonitor_r13_Type v_Mpdcch_NarrowbandsToMonitor;
    if (p_SysinfoType == sysinfoAndBR) { // Initiatile parameters for CE Mode A or B
      select(p_DL_ChBandwidth) {
        case (n25) { //5MHz
          // 36.523-3 Table 7.3.3.9.1-2: Narrow band allocation per cell (5 MHz)
          v_Mpdcch_NarrowbandsToMonitor := {2}
        }
        case (n50) { //10MHz
          //36.523-3 Table 7.3.3.9.1-4: Narrow band allocation per cell (10 MHz)
          v_Mpdcch_NarrowbandsToMonitor := {2}
          /*select (p_CellId) {
            case (eutra_Cell1, eutra_Cell2) { v_Mpdcch_NarrowbandsToMonitor := {1} }
            case (eutra_Cell4) { v_Mpdcch_NarrowbandsToMonitor := {2} }
            case else { FatalError(__FILE__, __LINE__, "invalid cellId for eMTC");}
          }*/
        }
        case (n100) {//20MHz
          //36.523-3 Table 7.3.3.9.1-6: Narrow band allocation per cell (20 MHz)
          v_Mpdcch_NarrowbandsToMonitor := {7}
        }
        case else { FatalError(__FILE__, __LINE__, "invalid value for ChannelBandwidth");}
      }
    }
    return v_Mpdcch_NarrowbandsToMonitor;
  }
...

Change 1.4 – Correction to MPDCCH configuration
	Template name
	1.   cs_MeasSubframePattern_r10_FDD ()
2.   cs_508_MPDCCH_Config_r13_Def ()

	Reason for change
	1.   The configured subframe pattern of MPDCCH is wrong. It configures 40 times 0, which means that the MPDCCH is not allowed in any subframes.
2.   The numberPRB_Pairs_r11 is set to n4 in the resourceBlockAssignment_r11, but if the numberPRB-Pairs-v1310 is configured (n6), the numberPRB_Pairs_r11 should be set to n2, see also TS 36.213 clause 9.1.4.4. and so the resourceBlockAssignment_r11 value needs to be also aligned accordingly.

	Summary of change
	1.   Modify the subframe pattern configuration to a meaningful value, e.g. avoiding subframes 0, 1, 4 and 5 used for SIBs / SIs, but allowing all other subframes, so that they would be scheduled on the same NB.
2.   Values are corrected according to the Core Spec reference.

	TTCN module
	Common\EUTRA\EUTRA_PdcchConfig.ttcn

	MCC160 Comment
	


Before change

...

  template (value) MeasSubframePattern_r10 cs_MeasSubframePattern_r10_FDD :=
  { /* @status    APPROVED (IMS, IMS_IRAT, LTE, LTE_A_IRAT, LTE_A_R10_R11, LTE_A_R12, LTE_IRAT, POS) */
    subframePatternFDD_r10 := '0000000000000000000000000000000000000000'B
  };
...

  template (value) EPDCCH_Config_r11  cs_508_MPDCCH_Config_r13_Def(template (value) MeasSubframePattern_r10 p_MeasSubframePattern_r10,
                                                                   template (value) EDPCCH_Mpdcch_StartSF_UESS_r13_Type p_Mpdcch_StartSF_UESS_r13,
                                                                   template (value) EDPCCH_Mpdcch_Narrowband_r13_Type   p_Mpdcch_Narrowband_r13) :=
  { //36.508 Table 4.6.3-2B: EPDCCH-Config-r11-DEFAULT cond CEmodeA FDD
    /* @status    APPROVED (IMS, IMS_IRAT, LTE, LTE_A_IRAT, LTE_A_R10_R11, LTE_A_R12, LTE_IRAT, POS) */
    config_r11 := {
      setup := {
        subframePatternConfig_r11 := {
         setup := {
            subframePattern_r11 :=  p_MeasSubframePattern_r10
          }
        },
        startSymbol_r11 := omit,
        setConfigToReleaseList_r11 := omit,
        setConfigToAddModList_r11  := {
          {
            setConfigId_r11  := 0,
            transmissionType_r11 := distributed,
            resourceBlockAssignment_r11 := {
              numberPRB_Pairs_r11   := n4,
              resourceBlockAssignment_r11 := '101001000101101110'B
            },
            dmrs_ScramblingSequenceInt_r11 := 0,
            pucch_ResourceStartOffset_r11 := 0,
            re_MappingQCL_ConfigId_r11    :=  omit,
            csi_RS_ConfigZPId2_r12 := omit,
            numberPRB_Pairs_v1310 := {setup := n6},
            mpdcch_config_r13 := {
             setup := {
                csi_NumRepetitionCE_r13 := sf1,
                mpdcch_pdsch_HoppingConfig_r13 := off,
                mpdcch_StartSF_UESS_r13 := p_Mpdcch_StartSF_UESS_r13,
                mpdcch_NumRepetition_r13 := r1, //@sic R5-169087 sic@
                mpdcch_Narrowband_r13 := p_Mpdcch_Narrowband_r13
            }
          }
        }}
      }
    }
  }
...

After change

...

  template (value) MeasSubframePattern_r10 cs_MeasSubframePattern_r10_FDD :=
  { /* @status    APPROVED (IMS, IMS_IRAT, LTE, LTE_A_IRAT, LTE_A_R10_R11, LTE_A_R12, LTE_IRAT, POS) */
    subframePatternFDD_r10 := '0011001111001100111100110011110011001111'B
  };

...

  template (value) EPDCCH_Config_r11  cs_508_MPDCCH_Config_r13_Def(template (value) MeasSubframePattern_r10 p_MeasSubframePattern_r10,
                                                                   template (value) EDPCCH_Mpdcch_StartSF_UESS_r13_Type p_Mpdcch_StartSF_UESS_r13,
                                                                   template (value) EDPCCH_Mpdcch_Narrowband_r13_Type   p_Mpdcch_Narrowband_r13) :=
  { //36.508 Table 4.6.3-2B: EPDCCH-Config-r11-DEFAULT cond CEmodeA FDD
    /* @status    APPROVED (IMS, IMS_IRAT, LTE, LTE_A_IRAT, LTE_A_R10_R11, LTE_A_R12, LTE_IRAT, POS) */
    config_r11 := {
      setup := {
        subframePatternConfig_r11 := {
         setup := {
            subframePattern_r11 :=  p_MeasSubframePattern_r10
          }
        },
        startSymbol_r11 := omit,
        setConfigToReleaseList_r11 := omit,
        setConfigToAddModList_r11  := {
          {
            setConfigId_r11  := 0,
            transmissionType_r11 := distributed,
            resourceBlockAssignment_r11 := {
              numberPRB_Pairs_r11   := n2,
              resourceBlockAssignment_r11 := '1110'B
            },
            dmrs_ScramblingSequenceInt_r11 := 0,
            pucch_ResourceStartOffset_r11 := 0,
            re_MappingQCL_ConfigId_r11    :=  omit,
            csi_RS_ConfigZPId2_r12 := omit,
            numberPRB_Pairs_v1310 := {setup := n6},
            mpdcch_config_r13 := {
             setup := {
                csi_NumRepetitionCE_r13 := sf1,
                mpdcch_pdsch_HoppingConfig_r13 := off,
                mpdcch_StartSF_UESS_r13 := p_Mpdcch_StartSF_UESS_r13,
                mpdcch_NumRepetition_r13 := r1, //@sic R5-169087 sic@
                mpdcch_Narrowband_r13 := p_Mpdcch_Narrowband_r13
            }
          }
        }}
      }
    }
  }
...

Change 1.5 – Correction to DciDlInfo
	Function name
	fl_DciDlInfo_Init ()

	Reason for change
	1.   The eMTC DL scheduling also overwrites the legacy BCCH settings shifting the SIB1 and SIs to the configured eMTC narrowband resulting in RB collisions with eMTC data.
2.   The DCISubframeRepetitionNumber = 1 is also wrong for the C-RNTI based scheduling, since the configured max. repetition number is r1.

	Summary of change
	1.   The case e_DlSchedulingClass_BCCH needs to be extended to shift the legacy BCCH back to the 0-3 RB range.
2.   Value set to 0.

	TTCN module
	Common\EUTRA_Templates\EUTRA_CellCfg_Templates.ttcn

	MCC160 Comment
	


Before change

...

  function fl_DciDlInfo_Init(DlSchedulingClass_Type p_SchedulClass,
                             Dl_Bandwidth_Type p_Dl_Bandwidth := n25,
                             EUTRA_CellId_Type p_CellId := eutra_Cell_NonSpecific)       //@sic R5s110034: delete sysinfoCombination parameter sic@
    return template (value) DciDlInfoCommon_Type
  {
    var template (value) DciDlInfoCommon_Type v_DciDlInfoCommon := cs_DciDlInfoCommon_Init;
    var EUTRA_ResourceAllocationScheme_Type v_ResourceAllocationScheme := fl_GetResourceAllocationScheme();  // @sic R5s110122 sic@
    var charstring v_TestcaseName := testcasename();
    var boolean v_TestcaseAttrib_3IntraFreq := f_GetTestcaseAttrib_3IntraFreq(v_TestcaseName);
    var boolean v_BLOrCE := pc_ue_CategoryDL_M1 or f_GetTestcaseAttrib_EnhancedCoverage(testcasename());
...

    // @sic R5-167569 sic@
    if (v_BLOrCE) {  // Overwrite
      //UE support category M1 or in enhanced coverage TC
      v_DciDlInfoCommon := cs_DciDlInfoCommon_Init;
      v_DciDlInfoCommon.RepetitionNumber := 0;
      v_DciDlInfoCommon.DCISubframeRepetitionNumber := 0;
      select (p_Dl_Bandwidth) {
        case (n25) {   // 5 Mhz
          v_DciDlInfoCommon.FreqDomainSchedul := cs_FreqDomainSchedul(19, 6);
        }
        case (n50) {  //10 Mhz
          v_DciDlInfoCommon.FreqDomainSchedul := cs_FreqDomainSchedul(37, 6);
          if (p_CellId == eutra_Cell4)
            {
              v_DciDlInfoCommon.FreqDomainSchedul := cs_FreqDomainSchedul(43, 6);
            }
        }
        case (n100) { //20 MhZ
          v_DciDlInfoCommon.FreqDomainSchedul := cs_FreqDomainSchedul(86, 6);
        }
      }
      select (p_SchedulClass) {
        case (e_DlSchedulingClass_BCCH)
          {
            // By default normal Sys info and Sys info BR is transmitted. for Sys Info BR there is no DCI on MPDCCH sent
          }
        case (e_DlSchedulingClass_PCCH)
          {
            v_DciDlInfoCommon.Format := dci_6_2;
          }
        case (e_DlSchedulingClass_RAR)
          {
            if (f_GetTestcaseAttrib_CEModeB(testcasename())) {v_DciDlInfoCommon.Format := dci_6_1B;}
            else {v_DciDlInfoCommon.Format := dci_6_1A;}
          }
        case else{ //C-RNTI based scheduling
          if (f_GetTestcaseAttrib_CEModeB(testcasename())) {v_DciDlInfoCommon.Format := dci_6_1B;}
          else {
            v_DciDlInfoCommon.Format := dci_6_1A;
            v_DciDlInfoCommon.DCISubframeRepetitionNumber := 1; //r2
          }
        }
      }
    }
    return v_DciDlInfoCommon;
  }
...

After change

...

  function fl_DciDlInfo_Init(DlSchedulingClass_Type p_SchedulClass,
                             Dl_Bandwidth_Type p_Dl_Bandwidth := n25,
                             EUTRA_CellId_Type p_CellId := eutra_Cell_NonSpecific)       //@sic R5s110034: delete sysinfoCombination parameter sic@
    return template (value) DciDlInfoCommon_Type
  {
    var template (value) DciDlInfoCommon_Type v_DciDlInfoCommon := cs_DciDlInfoCommon_Init;
    var EUTRA_ResourceAllocationScheme_Type v_ResourceAllocationScheme := fl_GetResourceAllocationScheme();  // @sic R5s110122 sic@
    var charstring v_TestcaseName := testcasename();
    var boolean v_TestcaseAttrib_3IntraFreq := f_GetTestcaseAttrib_3IntraFreq(v_TestcaseName);
    var boolean v_BLOrCE := pc_ue_CategoryDL_M1 or f_GetTestcaseAttrib_EnhancedCoverage(testcasename());
...

    // @sic R5-167569 sic@
    if (v_BLOrCE) {  // Overwrite
      //UE support category M1 or in enhanced coverage TC
      v_DciDlInfoCommon := cs_DciDlInfoCommon_Init;
      v_DciDlInfoCommon.RepetitionNumber := 0;
      v_DciDlInfoCommon.DCISubframeRepetitionNumber := 0;
      select (p_Dl_Bandwidth) {
        case (n25) {   // 5 Mhz
          v_DciDlInfoCommon.FreqDomainSchedul := cs_FreqDomainSchedul(19, 6);
        }
        case (n50) {  //10 Mhz
          v_DciDlInfoCommon.FreqDomainSchedul := cs_FreqDomainSchedul(37, 6);
          if (p_CellId == eutra_Cell4)
            {
              v_DciDlInfoCommon.FreqDomainSchedul := cs_FreqDomainSchedul(43, 6);
            }
        }
        case (n100) { //20 MhZ
          v_DciDlInfoCommon.FreqDomainSchedul := cs_FreqDomainSchedul(86, 6);
        }
      }
      select (p_SchedulClass) {
        case (e_DlSchedulingClass_BCCH)
          {
            // By default normal Sys info and Sys info BR is transmitted. for Sys Info BR there is no DCI on MPDCCH sent
            v_DciDlInfoCommon.FreqDomainSchedul := cs_FreqDomainSchedul(0, 4);
          }
        case (e_DlSchedulingClass_PCCH)
          {
            v_DciDlInfoCommon.Format := dci_6_2;
          }
        case (e_DlSchedulingClass_RAR)
          {
            if (f_GetTestcaseAttrib_CEModeB(testcasename())) {v_DciDlInfoCommon.Format := dci_6_1B;}
            else {v_DciDlInfoCommon.Format := dci_6_1A;}
          }
        case else{ //C-RNTI based scheduling
          if (f_GetTestcaseAttrib_CEModeB(testcasename())) {v_DciDlInfoCommon.Format := dci_6_1B;}
          else {
            v_DciDlInfoCommon.Format := dci_6_1A;
            //v_DciDlInfoCommon.DCISubframeRepetitionNumber := 1; //r2
          }
        }
      }
    }
    return v_DciDlInfoCommon;
  }
...

Change 1.6 – Correction to CQI Report Mode Aperiodic
	Template name
	cs_508_CQI_ReportConfig_Default ()

	Reason for change
	The CQI Report Mode Aperiodic value rm30 is not allowed for eMTC as per TS 36.213 clause 7.2.1.

	Summary of change
	Value corrected to rm20 for eMTC

	TTCN module
	Common\EUTRA_Templates\EUTRA_RRC_Templates.ttcn

	MCC160 Comment
	


Before change

...

module EUTRA_RRC_Templates {
  import from EUTRA_RRC_ASN1_Definitions language "ASN.1:2002" all;
  import from EUTRA_CommonDefs all;
  import from EUTRA_NB_CommonDefs all;
  import from EUTRA_DRB_Templates all;
  import from EUTRA_AuxiliaryFunctions all;
  import from EUTRA_Parameters all;
  import from CommonDefs all;
  import from Parameters all;
  import from EUTRA_SysInfo_Templates all;
  import from EUTRA_NB_SecurityFunctions all;
  import from EUTRA_NB_RRC_Templates all;
...

  template (value) CQI_ReportConfig cs_508_CQI_ReportConfig_Default(template (omit) CQI_ReportPeriodic p_CQI_ReportingPeriodic) :=
  { /* Default values according to 36.508 cl. 4.6.3 Table 4.6.3-2 */
    /* @status    APPROVED (IMS, IMS_IRAT, LTE, LTE_A_IRAT, LTE_A_R10_R11, LTE_A_R12, LTE_IRAT, POS) */
    cqi_ReportModeAperiodic := rm30,
    nomPDSCH_RS_EPRE_Offset := 0,
    cqi_ReportPeriodic := p_CQI_ReportingPeriodic
  };
...

After change

...

module EUTRA_RRC_Templates {
  import from EUTRA_RRC_ASN1_Definitions language "ASN.1:2002" all;
  import from EUTRA_CommonDefs all;
  import from EUTRA_NB_CommonDefs all;
  import from EUTRA_DRB_Templates all;
  import from EUTRA_AuxiliaryFunctions all;
  import from EUTRA_Parameters all;
  import from CommonDefs all;
  import from Parameters all;
  import from EUTRA_SysInfo_Templates all;
  import from EUTRA_NB_SecurityFunctions all;
  import from EUTRA_NB_RRC_Templates all;
  import from TestcaseProperties all;
...

  template (value) CQI_ReportConfig cs_508_CQI_ReportConfig_Default(template (omit) CQI_ReportPeriodic p_CQI_ReportingPeriodic) :=
  { /* Default values according to 36.508 cl. 4.6.3 Table 4.6.3-2 */
    /* @status    APPROVED (IMS, IMS_IRAT, LTE, LTE_A_IRAT, LTE_A_R10_R11, LTE_A_R12, LTE_IRAT, POS) */
    cqi_ReportModeAperiodic := f_EUTRA_CQI_ReportModeAperiodic(pc_ue_CategoryDL_M1 or f_GetTestcaseAttrib_EnhancedCoverage(testcasename())),
    nomPDSCH_RS_EPRE_Offset := 0,
    cqi_ReportPeriodic := p_CQI_ReportingPeriodic
  };
...

function f_EUTRA_CQI_ReportModeAperiodic(boolean p_IsEmtc := false) return template (omit) CQI_ReportModeAperiodic
  {
    var template (omit) CQI_ReportModeAperiodic v_CQI_ReportModeAperiodic := omit;
    if (p_IsEmtc) {
      v_CQI_ReportModeAperiodic := rm20;
    }
    else {
      v_CQI_ReportModeAperiodic := rm30;
    }    
    return v_CQI_ReportModeAperiodic;
  }
...

Change 1.7 – Correction to RACH-CE-LevelInfoList-r13
	Template name
	cs_RACH_CE_LevelInfoList ()

	Reason for change
	RACH-CE-LevelInfoList-r13 should include the same number of entries as the prach-ParametersListCE, see TS 36.331 clause 6.3.2.4. Entries, which have been configured originally, do not fit to CEmodeA.

	Summary of change
	List reduced to 2 elements.

	TTCN module
	Common\EUTRA_Templates\ EUTRA_SysInfo_Templates_BR.ttcn

	MCC160 Comment
	


Before change

...

  template(value) RACH_CE_LevelInfoList_r13 cs_RACH_CE_LevelInfoList :=
  { // acc to 36.508 Table 4.6.3-12: RACH-ConfigCommon-DEFAULT
    /* @status    APPROVED (IMS, IMS_IRAT, LTE, LTE_A_IRAT, LTE_A_R10_R11, LTE_A_R12, LTE_IRAT, POS) */
    {
      preambleMappingInfo_r13 := {
        firstPreamble_r13 := 0,
        lastPreamble_r13 := 14
      },
      ra_ResponseWindowSize_r13 := sf20,
      mac_ContentionResolutionTimer_r13 := sf80,
      rar_HoppingConfig_r13 := off
    },
    {
      preambleMappingInfo_r13 := {
        firstPreamble_r13 := 15,
        lastPreamble_r13 := 29
      },
      ra_ResponseWindowSize_r13 := sf80,
      mac_ContentionResolutionTimer_r13 := sf120,
      rar_HoppingConfig_r13 := off
    },

    {

      preambleMappingInfo_r13 := {

        firstPreamble_r13 := 30,

        lastPreamble_r13 := 44

      },

      ra_ResponseWindowSize_r13 := sf180,

      mac_ContentionResolutionTimer_r13 := sf200,

      rar_HoppingConfig_r13 := off

    },

    {

      preambleMappingInfo_r13 := {

        firstPreamble_r13 := 45,

        lastPreamble_r13 :=  60

      },

      ra_ResponseWindowSize_r13 := sf320,

      mac_ContentionResolutionTimer_r13 := sf480,

      rar_HoppingConfig_r13 := off

    }
  };
...

After change

...

  template(value) RACH_CE_LevelInfoList_r13 cs_RACH_CE_LevelInfoList :=
  { // acc to 36.508 Table 4.6.3-12: RACH-ConfigCommon-DEFAULT
    /* @status    APPROVED (IMS, IMS_IRAT, LTE, LTE_A_IRAT, LTE_A_R10_R11, LTE_A_R12, LTE_IRAT, POS) */
    {
      preambleMappingInfo_r13 := {
        firstPreamble_r13 := 0,
        lastPreamble_r13 := 14
      },
      ra_ResponseWindowSize_r13 := sf20,
      mac_ContentionResolutionTimer_r13 := sf80,
      rar_HoppingConfig_r13 := off
    },
    {
      preambleMappingInfo_r13 := {
        firstPreamble_r13 := 15,
        lastPreamble_r13 := 29
      },
      ra_ResponseWindowSize_r13 := sf80,
      mac_ContentionResolutionTimer_r13 := sf120,
      rar_HoppingConfig_r13 := off
    }/*,
    {
      preambleMappingInfo_r13 := {
        firstPreamble_r13 := 30,
        lastPreamble_r13 := 44
      },
      ra_ResponseWindowSize_r13 := sf180,
      mac_ContentionResolutionTimer_r13 := sf200,
      rar_HoppingConfig_r13 := off
    },
    {
      preambleMappingInfo_r13 := {
        firstPreamble_r13 := 45,
        lastPreamble_r13 :=  60
      },
      ra_ResponseWindowSize_r13 := sf320,
      mac_ContentionResolutionTimer_r13 := sf480,
      rar_HoppingConfig_r13 := off
    }*/
  };
...

Change 1.8 – Correction to subframe timing
	Function name
	f_EUTRA_TimingInfoAdjust ()

	Reason for change
	According to TS 36.523-3, subframe 4 should be avoided for eMTC transmissions.

	Summary of change
	Using subframe 9 for timing of eMTC transmissions.

	TTCN module
	Common\EUTRA\EUTRA_Timing.ttcn

	MCC160 Comment
	


Before change

...

module EUTRA_Timing {
  import from CommonDefs all;
  import from EUTRA_RRC_ASN1_Definitions language "ASN.1:2002" all;
  import from EUTRA_CommonDefs all;
  import from EUTRA_NB_CommonDefs all;
  import from EUTRA_ASP_TypeDefs all;
  import from EUTRA_ASP_SrbDefs all;
  import from EUTRA_CellInfo all;
  import from EUTRA_Component all;
  import from EUTRA_SRB_Templates all;
  import from EUTRA_AspCommon_Templates all;
  import from EUTRA_NB_Timing all;
  import from EUTRA_RRC_Templates all; //@sic R5-155983, R5s150924 sic@
...

  function f_EUTRA_TimingInfoAdjust(SubFrameTiming_Type p_TimingInfo) return SubFrameTiming_Type
  { /* @sic R5s110474 sic@ */
    var SubFrameTiming_Type v_AdjustedTiming;
    var UInt_Type v_SFNCalculation;
    v_AdjustedTiming := p_TimingInfo;
    if (p_TimingInfo.Subframe.Number > tsc_EUTRA_SN4) {
      v_SFNCalculation := (v_AdjustedTiming.SFN.Number + 1); // @sic R5w160204r1 sic@
      v_AdjustedTiming.SFN.Number := v_SFNCalculation mod 1024;
      v_AdjustedTiming.HSFN.Number := (v_AdjustedTiming.HSFN.Number + (v_SFNCalculation / 1024)) mod 1024; // @sic R5w160204r1 sic@
    }
    v_AdjustedTiming.Subframe.Number := tsc_EUTRA_SN4;   // SN4 to serve FDD and TDD
    return v_AdjustedTiming;
  }
...

After change

...

module EUTRA_Timing {
  import from CommonDefs all;
  import from EUTRA_RRC_ASN1_Definitions language "ASN.1:2002" all;
  import from EUTRA_CommonDefs all;
  import from EUTRA_NB_CommonDefs all;
  import from EUTRA_ASP_TypeDefs all;
  import from EUTRA_ASP_SrbDefs all;
  import from EUTRA_CellInfo all;
  import from EUTRA_Component all;
  import from EUTRA_SRB_Templates all;
  import from EUTRA_AspCommon_Templates all;
  import from EUTRA_NB_Timing all;
  import from EUTRA_RRC_Templates all; //@sic R5-155983, R5s150924 sic@
  import from TestcaseProperties all;
  import from Parameters all;
...

  function f_EUTRA_TimingInfoAdjust(SubFrameTiming_Type p_TimingInfo) return SubFrameTiming_Type
  { /* @sic R5s110474 sic@ */
    var SubFrameTiming_Type v_AdjustedTiming;
    var UInt_Type v_SFNCalculation;
    v_AdjustedTiming := p_TimingInfo;
    if (p_TimingInfo.Subframe.Number > tsc_EUTRA_SN4) {
      v_SFNCalculation := (v_AdjustedTiming.SFN.Number + 1); // @sic R5w160204r1 sic@
      v_AdjustedTiming.SFN.Number := v_SFNCalculation mod 1024;
      v_AdjustedTiming.HSFN.Number := (v_AdjustedTiming.HSFN.Number + (v_SFNCalculation / 1024)) mod 1024; // @sic R5w160204r1 sic@
    }
    if (pc_ue_CategoryDL_M1 or f_GetTestcaseAttrib_EnhancedCoverage(testcasename())) {
      v_AdjustedTiming.Subframe.Number := 9;
    }
    else {
      v_AdjustedTiming.Subframe.Number := tsc_EUTRA_SN4;   // SN4 to serve FDD and TDD
    }
    return v_AdjustedTiming;
  }
...

Change 2 – Correction specific to eMTC test case 6.1.2.2c
Change 2.1 – Correction to common + TC-specific functions in TTCN module ‘Idle_CellReselection_eMTC.ttcn’
	Function name
	Common:
f_EUTRA_CellInfo_SetSysInfoBR_Q_Rxlevmin ()    …  new function
f_EUTRA_CellInfo_SetSysInfo_Q_RxlevminCE

TC-specific:
f_TC_6_1_2_2c_EUTRA

	Reason for change
	1.   Function to set Q_Rxlevmin for SysInfoBR is missing. See associated Prose CR.
2.   The defined common function sets Q_RxlevminCE for SysInfo instead of SysInfoBR.

3.   The Q_RxLevMin values specified at the beginning of the test case body would allow to check the test purpose.
Moreover, SI’s instead of SI-BR are set in step 2.

	Summary of change
	1.   Added function f_EUTRA_CellInfo_SetSysInfoBR_Q_Rxlevmin. See associated Prose CR.
2.   Corrected the definition of common function f_EUTRA_CellInfo_SetSysInfoBR_Q_RxlevminCE.
3.   Corrected the Q_RxLevMin values at the beginning of the test case body acc. to the associated Prose CR (see attachment).  
Moreover, the function call in steps 2 was corrected to set SI-BR.

	TTCN module
	LTE_A_PRO\6\Idle_CellReselection_eMTC.ttcn

	MCC160 Comment
	


Before change

...

module Idle_CellReselection_eMTC
{
  import from EUTRA_RRC_ASN1_Definitions language "ASN.1:2002" all;
  import from EUTRA_Component all;
  import from EUTRA_CommonDefs all;
  import from CommonDefs all;
  import from EUTRA_NB_CommonDefs all;
  import from EUTRA_CellInfo all;
  import from EUTRA_ConfigurationSteps all;
  import from EUTRA_IdleMode all;
  import from EUTRA_NASSteps all;
  import from EUTRA_RRCSteps all;
  import from EUTRA_CommonProcedures all;
  import from EUTRA_InitialRegistration all;
  import from EUTRA_NASCommonFunctions all;
  import from EUTRA_RSRQ_Functions all;
  //=========================================================================
  // FUNCTIONS
  //=========================================================================
  /*
   * @desc      Assign the parameterised Qrxlevmin-CE value in SIB1 of the cell
   * @param     p_CellId
   * @param     p_Q_Rxlevmin
   */
  function f_EUTRA_CellInfo_SetSysInfo_Q_RxlevminCE(EUTRA_CellId_Type p_CellId,
                                                    Q_Rxlevmin_Type p_Q_Rxlevmin) runs on EUTRA_PTC
  {
    var template (value) EUTRA_CellInfo_Type v_CellInfo := f_EUTRA_CellInfo_Get(p_CellId);
      v_CellInfo.Sysinfo.BCCH_Info.SIB1.message_.c1.systemInformationBlockType1.nonCriticalExtension.nonCriticalExtension.nonCriticalExtension.nonCriticalExtension.nonCriticalExtension.cellSelectionInfoCE_r13.q_RxLevMinCE_r13 := p_Q_Rxlevmin;

    f_EUTRA_CellInfo_Set (p_CellId, v_CellInfo);
  }
  /*
   * @desc      Assign the parameterised Qqualmin-CE value in SIB1 of the cell
   * @param     p_CellId
   * @param     p_Q_Rxlevmin
   */
function f_EUTRA_CellInfo_SetSysInfo_Q_QualMinCE(EUTRA_CellId_Type p_CellId,
                                                    Q_QualMin_r9 p_Q_QualMinRSRQ) runs on EUTRA_PTC
  {
    var template (value) EUTRA_CellInfo_Type v_CellInfo := f_EUTRA_CellInfo_Get(p_CellId);
    v_CellInfo.Sysinfo.BCCH_Info.SIB1.message_.c1.systemInformationBlockType1.nonCriticalExtension.nonCriticalExtension.nonCriticalExtension.nonCriticalExtension.nonCriticalExtension.cellSelectionInfoCE_r13.q_QualMinRSRQ_CE_r13 := p_Q_QualMinRSRQ;
    f_EUTRA_CellInfo_Set (p_CellId, v_CellInfo);
  }
  /*
   * @desc      Assign the parameterised T_ReselectionEUTRA_CE value in SIB3 of the cell
   *            Should be moved to module EUTRA_Component
   * @param     p_CellId
   * @param     p_T_ReselectionEUTRA_CE
   */
  function f_EUTRA_CellInfo_SetSIB3_T_ReselectionEUTRA_CE(EUTRA_CellId_Type p_CellId,
                                                        T_ReselectionEUTRA_CE_r13  p_T_ReselectionEUTRA_CE) runs on EUTRA_PTC
  {
    var template (value) EUTRA_CellInfo_Type v_CellInfo := f_EUTRA_CellInfo_Get(p_CellId);
    v_CellInfo.Sysinfo.BCCH_Info.SIs[1].message_.c1.systemInformation.criticalExtensions.systemInformation_r8.sib_TypeAndInfo[0].sib3.t_ReselectionEUTRA_CE_r13 := p_T_ReselectionEUTRA_CE;
    f_EUTRA_CellInfo_Set (p_CellId, v_CellInfo);
    f_EUTRA_SetSIs_ValueTag(p_CellId);
  };
  //=========================================================================
  // TEST CASES
  //=========================================================================
  /*
   * @desc      REFERENCE TS 36.523-1 clause 6.1.2.2c
   */
  function f_TC_6_1_2_2c_EUTRA ( ) runs on EUTRA_PTC
  {
    var Q_RxLevMin v_Q_Rxlevmin := -41;
    var float v_TimerValue := 60.0;
    timer t_IdleMode_GenericTimer := tsc_IdleMode_GenericTimer;
    //Initialise all cells, security and mobile parameters
    f_EUTRA_Init(c2);
    //Set maximum cell powel level for Cell 1 to be used in creation
    f_EUTRA_CellInfo_InitMaxReferencePower (eutra_Cell1, -69);
    //Create and configure Cell 1
    f_EUTRA_CellConfig_Def (eutra_Cell1);
    //Perform Idle updated procedure to make sure UE is in a known state before test case starts
    f_EUTRA_Preamble (eutra_Cell1, STATE2_IDLEUPDATE);
    //Bring UE to initial state
    f_EUTRA_UE_Detach_SwitchOffUe (eutra_Cell1, RRC_IDLE);
    f_EUTRA_TestBody_Set (true);
    //@siclog "Step 1" siclog@
    //SS adjusts the cell levels according to row T1 of table 6.1.2.2c.3.2-1.
    f_EUTRA_SetCellPower (eutra_Cell1, -95);
    //@siclog "Step 2" siclog@
    f_EUTRA_CellInfo_SetSysInfo_Q_RxlevminCE (eutra_Cell1, v_Q_Rxlevmin);
    f_EUTRA_ModifySysinfoUE_Off (eutra_Cell1);
    //@siclog "Step 3" siclog@
    //UE is switched on
    f_EUTRA_SwitchOnUEandStartIP(eutra_Cell1);
    //@siclog "Step 4" siclog@
    //Check: Does the UE send an RRCConnectionRequest on Cell 1 within the next 60 s?
    if (f_EUTRA_RRC_RRCConnectionRequest_Check(eutra_Cell1, v_TimerValue)) {
      //* @verdict fail UE sent RRCConnectionRequest message on Cell 1 within v_TimerValue
      f_EUTRA_SetVerdictFailOrInconc (__FILE__, __LINE__, "Test Case 6.1.2.2c Step 4");
    }
    //@siclog "Step 5" siclog@
    //SS re-adjusts the cell-specific reference signal level of Cell 1 level according to row "T2" in table 6.1.2.2c.3.2-1.
    f_EUTRA_SetCellPower (eutra_Cell1, -69);
    t_IdleMode_GenericTimer.start;
    //@siclog "Step 6" siclog@
    //Check: Does the UE send an RRCConnectionRequest on Cell 1?
    f_EUTRA_RRC_ConnectionRequest_Def(eutra_Cell1);
    //Stop Idle Mode Geberic Timer
    t_IdleMode_GenericTimer.stop;
    f_EUTRA_PreliminaryPass (__FILE__, __LINE__, "Test Case 6.1.2.2c Step 6");
    //@siclog "Step 7 - 21" siclog@
    f_EUTRA_IdleUpdated_WithoutRRCConnReq (eutra_Cell1, STATE2_IDLEUPDATE);
    f_EUTRA_TestBody_Set (false);
    //Switch/power off UE
    f_EUTRA_Postamble (eutra_Cell1, E1_IDLE);
  };
...

After change

...

module Idle_CellReselection_eMTC
{
  import from EUTRA_RRC_ASN1_Definitions language "ASN.1:2002" all;
  import from EUTRA_Component all;
  import from EUTRA_CommonDefs all;
  import from CommonDefs all;
  import from EUTRA_NB_CommonDefs all;
  import from EUTRA_CellInfo all;
  import from EUTRA_ConfigurationSteps all;
  import from EUTRA_IdleMode all;
  import from EUTRA_NASSteps all;
  import from EUTRA_RRCSteps all;
  import from EUTRA_CommonProcedures all;
  import from EUTRA_InitialRegistration all;
  import from EUTRA_NASCommonFunctions all;
  import from EUTRA_RSRQ_Functions all;
  //=========================================================================
  // FUNCTIONS
  //=========================================================================
  /*
   * @desc      Assign the parameterised Qrxlevmin-CE value in SIB1 of the cell
   * @param     p_CellId
   * @param     p_Q_Rxlevmin
   */
   function f_EUTRA_CellInfo_SetSysInfoBR_Q_Rxlevmin(EUTRA_CellId_Type p_CellId,
                                                    Q_Rxlevmin_Type p_Q_Rxlevmin) runs on EUTRA_PTC
  {
    var template (value) EUTRA_CellInfo_Type v_CellInfo := f_EUTRA_CellInfo_Get(p_CellId);
    v_CellInfo.Sysinfo.BCCH_BR_Info.SIB1.message_.c1.systemInformationBlockType1_BR_r13.cellSelectionInfo.q_RxLevMin := p_Q_Rxlevmin;
    f_EUTRA_CellInfo_Set (p_CellId, v_CellInfo);
  }
  function f_EUTRA_CellInfo_SetSysInfoBR_Q_RxlevminCE(EUTRA_CellId_Type p_CellId,
                                                    Q_Rxlevmin_Type p_Q_Rxlevmin) runs on EUTRA_PTC
  {
    var template (value) EUTRA_CellInfo_Type v_CellInfo := f_EUTRA_CellInfo_Get(p_CellId);
         v_CellInfo.Sysinfo.BCCH_BR_Info.SIB1.message_.c1.systemInformationBlockType1_BR_r13.nonCriticalExtension.nonCriticalExtension.nonCriticalExtension.nonCriticalExtension.nonCriticalExtension.cellSelectionInfoCE_r13.q_RxLevMinCE_r13 := p_Q_Rxlevmin;
    f_EUTRA_CellInfo_Set (p_CellId, v_CellInfo);
  }
  /*
   * @desc      Assign the parameterised Qqualmin-CE value in SIB1 of the cell
   * @param     p_CellId
   * @param     p_Q_Rxlevmin
   */
  function f_EUTRA_CellInfo_SetSysInfo_Q_QualMinCE(EUTRA_CellId_Type p_CellId,
                                                    Q_QualMin_r9 p_Q_QualMinRSRQ) runs on EUTRA_PTC
  {
    var template (value) EUTRA_CellInfo_Type v_CellInfo := f_EUTRA_CellInfo_Get(p_CellId);
    v_CellInfo.Sysinfo.BCCH_Info.SIB1.message_.c1.systemInformationBlockType1.nonCriticalExtension.nonCriticalExtension.nonCriticalExtension.nonCriticalExtension.nonCriticalExtension.cellSelectionInfoCE_r13.q_QualMinRSRQ_CE_r13 := p_Q_QualMinRSRQ;
    f_EUTRA_CellInfo_Set (p_CellId, v_CellInfo);
  }
  /*
   * @desc      Assign the parameterised T_ReselectionEUTRA_CE value in SIB3 of the cell
   *            Should be moved to module EUTRA_Component
   * @param     p_CellId
   * @param     p_T_ReselectionEUTRA_CE
   */
  function f_EUTRA_CellInfo_SetSIB3_T_ReselectionEUTRA_CE(EUTRA_CellId_Type p_CellId,
                                                        T_ReselectionEUTRA_CE_r13  p_T_ReselectionEUTRA_CE) runs on EUTRA_PTC
  {
    var template (value) EUTRA_CellInfo_Type v_CellInfo := f_EUTRA_CellInfo_Get(p_CellId);
    v_CellInfo.Sysinfo.BCCH_Info.SIs[1].message_.c1.systemInformation.criticalExtensions.systemInformation_r8.sib_TypeAndInfo[0].sib3.t_ReselectionEUTRA_CE_r13 := p_T_ReselectionEUTRA_CE;
    f_EUTRA_CellInfo_Set (p_CellId, v_CellInfo);
    f_EUTRA_SetSIs_ValueTag(p_CellId);
  };
  //=========================================================================
  // TEST CASES
  //=========================================================================
  /*
   * @desc      REFERENCE TS 36.523-1 clause 6.1.2.2c
   */
  function f_TC_6_1_2_2c_EUTRA ( ) runs on EUTRA_PTC
  {
    //var Q_RxLevMin v_Q_Rxlevmin := -41;

    var Q_RxLevMin v_Q_Rxlevmin := -27;
    var Q_RxLevMin v_Q_Rxlevmin_CE := -41;
    var float v_TimerValue := 60.0;
    timer t_IdleMode_GenericTimer := tsc_IdleMode_GenericTimer;
    //Initialise all cells, security and mobile parameters
    f_EUTRA_Init(c2);
    //Set maximum cell powel level for Cell 1 to be used in creation
    f_EUTRA_CellInfo_InitMaxReferencePower (eutra_Cell1, -69);
    //Create and configure Cell 1
    f_EUTRA_CellConfig_Def (eutra_Cell1);
    //Perform Idle updated procedure to make sure UE is in a known state before test case starts
    f_EUTRA_Preamble (eutra_Cell1, STATE2_IDLEUPDATE);
    //Bring UE to initial state
    f_EUTRA_UE_Detach_SwitchOffUe (eutra_Cell1, RRC_IDLE);
    f_EUTRA_TestBody_Set (true);
    //@siclog "Step 1" siclog@
    //SS adjusts the cell levels according to row T1 of table 6.1.2.2c.3.2-1.
    f_EUTRA_SetCellPower (eutra_Cell1, -95);
    //@siclog "Step 2" siclog@
    //f_EUTRA_CellInfo_SetSysInfo_Q_RxlevminCE (eutra_Cell1, v_Q_Rxlevmin);
    f_EUTRA_CellInfo_SetSysInfoBR_Q_Rxlevmin(eutra_Cell1, v_Q_Rxlevmin);
    f_EUTRA_CellInfo_SetSysInfoBR_Q_RxlevminCE (eutra_Cell1, v_Q_Rxlevmin_CE);
    //f_EUTRA_ModifySysinfoUE_Off (eutra_Cell1);
    //The function modifies SIB1-BR message also
    f_EUTRA_ModifySysinfoUE_Off (eutra_Cell1,-,true);
    //@siclog "Step 3" siclog@
    //UE is switched on
    f_EUTRA_SwitchOnUEandStartIP(eutra_Cell1);
    //@siclog "Step 4" siclog@
    //Check: Does the UE send an RRCConnectionRequest on Cell 1 within the next 60 s?
    if (f_EUTRA_RRC_RRCConnectionRequest_Check(eutra_Cell1, v_TimerValue)) {
      //* @verdict fail UE sent RRCConnectionRequest message on Cell 1 within v_TimerValue
      f_EUTRA_SetVerdictFailOrInconc (__FILE__, __LINE__, "Test Case 6.1.2.2c Step 4");
    }
    //@siclog "Step 5" siclog@
    //SS re-adjusts the cell-specific reference signal level of Cell 1 level according to row "T2" in table 6.1.2.2c.3.2-1.
    f_EUTRA_SetCellPower (eutra_Cell1, -69);
    t_IdleMode_GenericTimer.start;
    //@siclog "Step 6" siclog@
    //Check: Does the UE send an RRCConnectionRequest on Cell 1?
    f_EUTRA_RRC_ConnectionRequest_Def(eutra_Cell1);
    //Stop Idle Mode Geberic Timer
    t_IdleMode_GenericTimer.stop;
    f_EUTRA_PreliminaryPass (__FILE__, __LINE__, "Test Case 6.1.2.2c Step 6");
    //@siclog "Step 7 - 21" siclog@
    f_EUTRA_IdleUpdated_WithoutRRCConnReq (eutra_Cell1, STATE2_IDLEUPDATE);
    f_EUTRA_TestBody_Set (false);
    //Switch/power off UE
    f_EUTRA_Postamble (eutra_Cell1, E1_IDLE);
  };
...

Change 2.2 – Correction to common functions in TTCN module ‘EUTRA_RRCSteps.ttcn’
	Function name
	f_EUTRA_ModifySysinfoUE_Off

	Reason for change
	Function defined in a way that SIBs cannot be modified independently from SIB1-BR. 

	Summary of change
	Extended the function definition.

	TTCN module
	Common\EUTRA\EUTRA_RRCSteps.ttcn

	MCC160 Comment
	


Before change

...

  function f_EUTRA_ModifySysinfoUE_Off(EUTRA_CellId_Type p_CellId,
                                       template (omit) Null_Type p_StopSib1Transmission := omit) //@sic R5-142409 sic@
    runs on EUTRA_PTC                         
  { // @sic R5s100008 R5-157569 sic@
    var SubFrameTiming_Type     v_Timing, v_Timing_SFN;
    var EUTRA_CellSysInfo_Type  v_Sysinfo:= f_EUTRA_CellInfo_GetSYSINFO (p_CellId);
    //Get the current SFN and subframe
    v_Timing := f_EUTRA_GetCurrentTiming(p_CellId);
    //Modify sysinfo now as UE is OFF
    v_Timing_SFN := v_Timing;
    v_Timing_SFN.HSFN.Number := (v_Timing.HSFN.Number + (v_Timing.SFN.Number + tsc_SfnDelay)/1024) mod 1024;
    v_Timing_SFN.SFN.Number := (v_Timing.SFN.Number + tsc_SfnDelay) mod 1024;
    f_EUTRA_SS_ConfigureSysinfo (p_CellId, cs_TimingInfo_SFN(v_Timing_SFN.SFN.Number, v_Timing_SFN.HSFN.Number), v_Sysinfo.SB1_ValueTagChanged, p_StopSib1Transmission);
    //Wait the duration of the delay to ensure new SYSINFO takes effect
    f_Delay (int2float(tsc_SfnDelay)/100.0);
  }

...

After change

...

  function f_EUTRA_ModifySysinfoUE_Off(EUTRA_CellId_Type p_CellId,
                                       template (omit) Null_Type p_StopSib1Transmission := omit, boolean p_ModifySIB1_BROnly := false) //@sic R5-142409 sic@
    runs on EUTRA_PTC                         
  { // @sic R5s100008 R5-157569 sic@
    var SubFrameTiming_Type     v_Timing, v_Timing_SFN;
    var EUTRA_CellSysInfo_Type  v_Sysinfo:= f_EUTRA_CellInfo_GetSYSINFO (p_CellId);
    //Get the current SFN and subframe
    v_Timing := f_EUTRA_GetCurrentTiming(p_CellId);
    //Modify sysinfo now as UE is OFF
    v_Timing_SFN := v_Timing;
    v_Timing_SFN.HSFN.Number := (v_Timing.HSFN.Number + (v_Timing.SFN.Number + tsc_SfnDelay)/1024) mod 1024;
    v_Timing_SFN.SFN.Number := (v_Timing.SFN.Number + tsc_SfnDelay) mod 1024;
    f_EUTRA_SS_ConfigureSysinfo (p_CellId, cs_TimingInfo_SFN(v_Timing_SFN.SFN.Number, v_Timing_SFN.HSFN.Number), v_Sysinfo.SB1_ValueTagChanged, p_StopSib1Transmission);
    //Wait the duration of the delay to ensure new SYSINFO takes effect
    f_Delay (int2float(tsc_SfnDelay)/100.0);
  }

...

5. Execution Log Files

5.1 Altair ALT1210
The Altair ALT1210 UE passed this LTE-A-PRO (Rel-13) eMTC test case on R&S CMW500 Multi-RAT Protocol Tester in LTE FDD band 13. The documentation below is enclosed as evidence of the successful test case run [1]:

· Test Case Execution log file
Rohde-Schwarz\tc_6_1_2_2c_LTE_FDD_Altair.log   
(Note: PICS/PIXIT settings are captured at the beginning of the TLI log)


In the log file (in .txt / -html format) the complete test case execution can be seen. All TLI events are presented and message contents are fully decoded and can be verified. Preliminary verdicts and the final test case verdict can be seen in the log file.

5.2 Qualcomm MDM9607
The Qualcomm MDM 9607 UE passed this LTE-A-PRO (Rel-13) eMTC test case on R&S CMW500 Multi-RAT Protocol Tester in LTE FDD band 4 The documentation below is enclosed as evidence of the successful test case run [1]:

· Test Case Execution log file
Rohde-Schwarz\tc_6_1_2_2c_LTE_FDD_Qualcomm.log   
(Note: PICS/PIXIT settings are captured at the beginning of the TLI log)


In the log file (in .txt / -html format) the complete test case execution can be seen. All TLI events are presented and message contents are fully decoded and can be verified. Preliminary verdicts and the final test case verdict can be seen in the log file.

5.3 Sequans Monarch LTE Cat M1
The Sequans Monarch LTE Cat M1 UE passed this LTE-A-PRO (Rel-13) eMTC test case on R&S CMW500 Multi-RAT Protocol Tester in LTE FDD band 13. The documentation below is enclosed as evidence of the successful test case run [1]:

· Test Case Execution log file
Rohde-Schwarz\tc_6_1_2_2c_LTE_FDD_Sequans.log   
(Note: PICS/PIXIT settings are captured at the beginning of the TLI log)


In the log file (in .txt / -html format) the complete test case execution can be seen. All TLI events are presented and message contents are fully decoded and can be verified. Preliminary verdicts and the final test case verdict can be seen in the log file.

